UNIVERSITÉ DE POITIERS **2008-2009**
Master de Mathématiques et Applications
Mention Modélisation Mathématique et Analyse Statistique
1M10

Correction du TP 3

Exercice 1 Soit $f \in L^1(\mathbb{R}, \mathbb{C})$. Pour $\xi \in \mathbb{R}$, et $T > 0$, on a

$$\left| \hat{f}(\xi) - \int_{-T}^{T} f(x) e^{-i\xi x} dx \right| \leq \int_{|x| \geq T} |f(x)| dx,$$

donc, par convergence dominée,

$$\sup_{\xi \in \mathbb{R}} \left| \hat{f}(\xi) - \int_{-T}^{T} f(x) e^{-i\xi x} dx \right| \to 0$$

lorsque $T \to +\infty$.

Soit $T > 0$ fixé et $g \in C^1([-T, T], \mathbb{C})$. Pour $N \in \mathbb{N}^\star$, on pose $h = 2T/N$ et $x_j = -T + jh$, $j = 0, 1, \ldots, N$. On écrit

$$\int_{-T}^{T} g(x) dx - h \sum_{j=0}^{N-1} g(x_j) = \sum_{j=0}^{N-1} \int_{x_j}^{x_{j+1}} [g(x) - g(x_j)] dx.$$

Par le théorème des accroissements finis,

$$\int_{x_j}^{x_{j+1}} |g(x) - g(x_j)| \leq \int_{x_j}^{x_{j+1}} (x - x_j) \sup_{s \in [x_j, x_{j+1}]} |g'(s)| = h^2/2 \sup_{s \in [x_j, x_{j+1}]} |g'(s)|,$$

on en déduit l'estimation d'erreur suivante pour la formule des rectangles composée :

$$| \int_{-T}^{T} g(x) dx - h \sum_{j=0}^{N-1} g(-T + jh) | \leq Th \sup_{x \in [-T, T]} |g'(x)|.$$

On peut vérifier avec le noyau de Péano que la constante trouvée est optimale.

On applique cette formule avec $g(x) = f(x) e^{-i\xi x}$ pour $f \in C^1([-T, T], \mathbb{C})$. On a $g'(x) = f'(x) e^{-i\xi x} - i\xi f(x) e^{-i\xi x}$, et donc

$$\left| \int_{-T}^{T} f(x) e^{-i\xi x} dx - h \sum_{j=0}^{N-1} f(x_j) e^{-i\xi x_j} \right| \leq Th \left(\|f'\|_{L^\infty(-T,T)} + |\xi| \|f\|_{L^\infty(-T,T)} \right). \tag{1}$$

Exercice 2 Pour $f(x) = 1$ si $|x| \leq 1$ et 0 sinon, on a

$$\hat{f}(\xi) = 2 \sin(\xi)/\xi, \quad \xi \in \mathbb{R}.$$

Il s'agit de la fonction sinus cardinal (sinc en SCILAB).

Exercice 3

```
function [hf,xi]=spectre(f,T,M)
  //etant donne une fonction f(x), x=[-T:T/M,T-T/M], avec M=2^m,
  //calcule le spectre approche aux points xi=kpi/T, k=-M:1:M-1
 x=[-T:T/M:T-T/M];
  k=[0:1:2*M-1];
  aux=(T/M)*cos(k*%pi).*fft(f(x),-1)
  hf=zeros(x);
  hf(1:M)=aux(M+1:2*M);
  hf(M+1:2*M)=aux(1:M);
 xi=[-M*%pi/T:%pi/T:M*%pi/T-%pi/T];
endfunction
```

Exercice 4 Programme : transformeedefourier.sce

```
  clear
  getf("spectre.sci")
  deff('y=f(x)','y=ones(x).*((x>-1)&(x<1))')
  deff('y=hfsol(xi)','y=2*sinc(xi)')

  T=2^3
  M=2^6

  [hf,xi]=spectre(f,T,M);
  clf
  plot2d(xi',[real(hf') hfsol(xi')],[-4 5],leg="disc@exact")
  titre=strcat(['T=',string(T),' | M=', string(M)]);
  xtitle(titre,'xi','hf')
```

Il y a une différence entre $\hat{f}$ et son calcul numérique à cause de l'approximation introduite par la formule des rectangles.

Un bon moyen de choisir M par rapport à T est $T = \sqrt{M}$, de sorte que T est grand et T/M petit. Remarquer cependant qu'avec ce choix, le produit $hT = 2T^2/M = 2$ dans l'estimation d'erreur (1).

Exercice 5 Programme : transformeedefourier2.sce

```
  clear
  getf("spectre.sci")
  deff('y=f(x)','y=exp(-(x+3).^2/2).*cos(30*x)+sqrt(50)*exp(-50*(x-8.5).^2).*sin(50*x)')

  T=10.24
  M=2^10
  x=[-T:T/M:T];
  clf
  subplot(2,1,1)
  plot2d(x,f(x))
// title('','x','f')
  [hf,xi]=spectre(f,T,M);
  subplot(2,1,2)
  plot2d(xi,real(hf),rect=[-100,-1.5,100,1.5])
//  plot2d(xi',[real(hf') hfsol(xi')],[-4 5],leg="disc@exact")
  titre=strcat(['T=',string(T),' | M=', string(M)]);
  xtitle(titre,'xi','hf')
```



Figure 1: Exécution de transformeedefourier.sce

On remarque que f est (à un sin et cos près), la somme de 2 signaux gaussiens. Or la transformée de Fourier d'un signal gaussien est un signal gaussien. On retrouve essentiellement ces deux signaux gaussiens sur la Figure 2.



Figure 2: Exécution de transformeedefourier2.sce